


## *Recursive Utilization of FINTG and FROOT.*

Like the original SOLVE and INTEG did, both **FROOT** & **FINTG** support “crossed” nested calls from one another, i.e. you can call FROOT from an integrand function being used by FINTG, and you can call FINTG in the root-finding function definition for FROOT. However, it is not possible to recursively call either one of these functions sequentially from within a FOCAL routine. Any attempts to do so triggers the “**RECURSION**” error message and the execution aborts.

The SIROM provides a set of MCODE functions and two FOCAL routines to overcome this limitation. Each time FROOT/FINTG is executed it creates a dedicated memory buffer to store the application data and to perform all the math. The basis of the operation is the use of a secondary memory area for the nested call of the function, not conflicting with the initial memory buffer created in the first call. The main loop uses the initial buffer #14, and the operand function in turn creates a secondary buffer #14 to use for the nested loop – deleting it after it’s complete.

In order to reuse the existing code, we’ll trick the OS changing the id# of the initial buffer #14 right before the second call – *not deleting it but cloaking it in the I/O Memory area of the calculator*. The operand function re-labels the buffer with id#13 (using function **CLOAK**), then the nested call to FROOT/INTEG creates and uses a new buffer #14 to perform its task, and deletes it upon completion – returning the execution to the “operand” function FOCAL routine. Before the execution is returned to the driver program, the cloaked buffer is re-issued as id#14 (using function **EXPOSE**) so things can be picked up exactly where there were left off before calling the nested subroutine.

If you must know, all **CLOAK** and **EXPOSE** do is changing the buffer id# of the initial buffer created in the first call to FROOT/INTEG - first from 14 to 13, and then back to 14. Prior to all this a third function (**RESET**) is used to check for pre-existing buffers with id#13 – deleting it if found.



### 2D Driver Routines and Rules of Engagement.

The main programs for double integrals and system of 2 equations are “**FITG2**” and “**FRT2**”. Each one has an auxiliary routine associated with it, which acts as the first level operand function and issues a second nested call for the integrand or the second equation appropriately, as follows:

For **FITG2**, the function name  $f(x,y)$  is expected in ALPHA, and the four integral limits in the stack in the pattern “ $y1, y2, x1, x2$ ” –  $(y1,y2)$  for the outer integral, and  $(x1,x2)$  for the inner one.

- *The integrand function is to be programmed assuming  $x$  is in R01, and  $y$  in the stack.*

For **FRT2**, both function names are expected to be in Alpha separated by comma (like “**F1,F2**”), and the guesses entered in the stack, with the pattern “ $x1, x2, y1, y2$ ” - with  $(x1, x2)$  for  $f1(x,y)$  and  $(y1, y2)$  for  $f2(x,y)$ .

- *The second operand function  $f2(x,y)$  is executed first. It assumes  $x$  in R01 and  $y$  in the stack.*
- *The first operand function  $f1(x,y)$  assumes  $x$  in R01 and  $y$  in R02.*
- *You decide which one is F1 and F2 by their order in the ALPHA string*

All buffer management is made automatically by the auxiliary routines “**\*2D**” and “**\*FG**”.

## Routine Listings.

Here are the routine listings for your perusal. Notably **FRT2** introduces more complexity to process the function names – entered as comma-separated strings in ALPHA – and due to the indirect call to  $f_1(x,y)$  at the end of the auxiliary routine **"\*FG"** - which is not required in the double integration case, as it's just one function involved. **CLAC** and **ASWAP** are borrowed from the ALPHA ROM – and need the Library#4 present in the calculator. They're only used for **FRT2**.

01	<b>LBL "FRT2"</b>		01	<b>LBL "FITG2"</b>	
02	<b>CLKEYS</b>	<i>no keys assigned</i>	02	<b>CLKEYS</b>	<i>no keys assigned</i>
03	<b>ASTO 00</b>	<i>save string</i>	03	<b>ASTO 00</b>	<i>save in R00</i>
04	<b>ASWAP</b>	<i>swap around ", "</i>	04	<b>STO 03</b>	<i>upper limit2</i>
05	<b>CLAC</b>	<i>remove second</i>	05	<b>RDN</b>	
06	<b>ASTO 05</b>	<i>save in R05</i>	06	<b>STO 02</b>	<i>lower limit2</i>
07	<b>CLA</b>		07	<b>RDN</b>	
08	<b>ARCL 00</b>	<i>recall string</i>	08	<b>RESET</b>	<i>reset buffers</i>
09	<b>CLAC</b>	<i>remove second</i>	09	<b>"2D"</b>	<i>first level operand</i>
10	<b>ASTO 00</b>	<i>save in R00</i>	10	<b>FINTG</b>	<i>call first round</i>
11	<b>STO 04</b>	<i>upper bound2</i>	11	<b>RTN</b>	<i>done.</i>
12	<b>RDN</b>		12	<b>LBL "*2D"</b>	
13	<b>STO 03</b>	<i>lower bound2</i>	13	<b>STO 01</b>	<i>Save x for later</i>
14	<b>RDN</b>		14	<b>CLOAK</b>	<i>mask buffer id#</i>
15	<b>RESET</b>	<i>reset buffers</i>	15	<b>RCL 02</b>	<i>lower limit2</i>
16	<b>"*FG"</b>	<i>first level operand</i>	16	<b>RCL 03</b>	<i>upper limit2</i>
17	<b>FROOT</b>	<i>call first round</i>	17	<b>CLA</b>	
18	<b>RCL 02</b>	<i>y solution</i>	18	<b>ARCL 00</b>	<i>f(x,y)</i>
19	<b>X&lt;&gt;Y</b>	<i>arrange in stack</i>	19	<b>FINTG</b>	<i>nested call</i>
20	<b>" -, "</b>	<i>appends</i>	20	<b>EXPOSE</b>	<i>re-issue buf id#</i>
21	<b>ARCL 00</b>	<i>f1(x,y) name</i>	21	<b>END</b>	<i>ready</i>
22	<b>ASWAP</b>	<i>swap around</i>			
23	<b>RTN</b>	<i>done(!)</i>			
24	<b>LBL "*FG"</b>				
25	<b>STO 01</b>	<i>save x for later</i>			
26	<b>CLOAK</b>	<i>mask buffer id#</i>			
27	<b>RCL 03</b>	<i>lower bound2</i>			
28	<b>RCL 04</b>	<i>upper bound2</i>			
29	<b>CLA</b>				
30	<b>ARCL 05</b>	<i>f2(x,y)</i>			
31	<b>FROOT</b>	<i>nested call</i>			
32	<b>EXPOSE</b>	<i>re-issue buf id#</i>			
33	<b>STO 02</b>	<i>Save yo result</i>			
34	<b>XEQ IND 00</b>	<i>calculates f1(x,Yo)</i>			
35	<b>END</b>				

**FINTG** uses registers {R00-R03} and leaves the results in X and R01. The function name is left in ALPHA (6-chars max).

**FRT2** uses registers {R00-R05} and leaves the results in the stack registers {X, Y} and {R01, R02} for the 2-equation roots. The comma-separated function names string is left in ALPHA (6-chars max for each name).

*Comments.*

The new functions to support the nested configuration are simplified versions of some general-purpose buffer utilities, available in other extension modules as follows:

- **RESET** is equivalent to the sequence { 13, **B?**, **CLB**, RDN }
- **CLOAK** is equivalent to the sequence { 14.013 , **REIDBF**, RDN }
- **EXPOSE** is equivalent to the sequence: { 13.014 , **REIDBF** , RDN }

**B?** and **CLB** are available in the OS/X ROM, and **REIDBF** in the RAMPAGE ROM.

Using the simplified versions is more intuitive for math-oriented users, and besides freed up some space for additional examples in the SIROM.

While you can use **RESET** at any time (which will delete buff #13 if present, or do nothing if not present), using **CLOAK** and **EXPOSE** will generally result in the error message "*BUF ERR*". They're meant to be used only while buffer #14 exists, which is tightly controlled by the code in FINTG and FROOT – and furthermore, the SIROM uses the I/O\_PAUSE interrupt as a "search & destroy" for buffer#14 at all times. Refer to the corresponding section in the **SandMath** manual to read more on this subject.

*Caveat emptor.*

There's a price to pay for this buffer trickery, and that's the loss of the USER key assignments. As you can see in the listings above, the main routines call **CLKEYS** to make the operation more reliable (this avoids spurious buffer errors due to memory overwrites). You can save them in an X-Mem file using **SAVEKA** and then recover them with **GETKA** after the fact (both functions are also available in the AMC\_OS/X ROM).

These routines are not fast, their interest is in the methodology - not optimized for speed to say the least. If you need faster responses, then the SandMath provides dedicated MCODE functions for many of these and yet some more.

Bear in mind that the INTEG-based method to define special functions is not an efficient one from the mathematical standpoint, but it is a godsend for engineering problems. Also FROOT is not perfect or fool-proof either, so choosing a good initial guess is of high importance. If FRT2 fails to find a root (in either variable), it'll present the error message "**NO ROOT**" – Change the limits and try again.

The following examples should provide a good overview into the details of the programming.

*Example 1.* Calculate the integral of the Bessel  $J_n$  function,  $ITJ(1,3) = \text{INT}(0,3) \{ J(1,t).dt \}$  using the integral definition as reference:

$$J_n(x) = \frac{1}{\pi} \int_0^\pi \cos(n\tau - x \sin \tau) d\tau.$$

Program Code is below. Note that you don't need to worry about the buffer management, that's done automatically by the driver routines all transparently to the user.

01	LBL "ITJB"		13	LBL " *JN"	inner variable t in stack
02	X<>Y	order n to X	14	RAD	angular mode
03	STO 04	order saved in R04	15	RCL 04	get order
04	CLX	lower outer limit	16	*	n.t
05	X<>Y	upper outer limit	17	X<>Y	inner variable t
06	0	lower inner limit	18	SIN	sin t
07	PI	upper inner limit	19	RCL 01	outer variable
08	"*JN"	function name	20	*	x.sin t
09	XROM " ITG2"	double integration	21	-	n.t - x.sin t
10	PI	adjust factor	22	COS	cos (n.t - x.sin t)
11	/	final result	23	END	integrand complete.
12	RTN	done.			

As mentioned before, *speed is not this method's forte*. Even on V41 in turbo mode it'll take a good 75 seconds to return 1.260052 (in FIX 6). This was not the goal of the example, but to clarify the general guidelines and showcase the conceptual approach. If you want a fast result you're encouraged to use **JBS** in the SandMath, or even better the **ITJ**(sub)function also in the SandMath, which uses the Generalized, Regularized Hypergeometric function for the calculation – a world of differences...


*Comment.* This particular example is of course much better dealt with using the well-known expression between the Bessel function  $J_1$  and  $J_0$  shown below (proving once again that it's always good to check your math before embarking in long and winding paths):

$$\int_0^x J_1(t) dt = 1 - J_0(x)$$

thus:



$$\int_0^3 J_1(t) dt = 1 - J_0(3) \approx 1.26005$$

Here's an interesting plot showing the integral function of  $J_1(x)$  between ]-15 . 15[



Example 2. Calculate the solution for the system of non-linear equations below:

$$\begin{cases} f1(x,y) = x - \sin(x + y) \\ f2(x,y) = y - \cos(x - y) \end{cases} \quad \text{Solution:} \quad \begin{aligned} x &= 0,935082064 \\ y &= 0,998020058 \end{aligned}$$



The equations are programmed as shown below. Note how the convention is observed, with the y value assumed in the stack for the second function and in R02 for the first one; whilst x is always assumed in R01 for both functions. The solutions are obtained in about 3 seconds (FIX 9) using V41 in Turbo mode.

ALPHA, "F1,F2", ALPHA, 1, ENTER^, 2, ENTER^, 1, ENTER^, 2, XEQ "FRT2"

01	<b>LBL "F1"</b>		09	<b>LBL "F2"</b>	
02	<b>RCL 01</b>	x	10	<b>RAD</b>	
03	<b>RCL 02</b>	y	11	<b>CHS</b>	-y
04	<b>+</b>	x+y	12	<b>RCL 01</b>	x
05	<b>SIN</b>	sin(x+y)	13	<b>+</b>	x-y
06	<b>RCL 01</b>	x	14	<b>COS</b>	cos(x-y)
07	<b>-</b>	sin(x+y)-x	15	<b>X&lt;&gt;Y</b>	y
08	<b>RTN</b>		16	<b>-</b>	cos(x-y)-y
			17	<b>END</b>	

### More Examples.

The table below lists the examples provided in the module, with their type of application and possible recursion/nested approaches.

Example	Description	Used Program	Application Type	Dimension
<b>2DF1</b>	Integrand example1	<b>2DITG</b>	2D Integration	Simple
<b>2DF2</b>	Integrand example2	<b>2DITG</b>	2D Integration	Simple
<b>2DF3</b>	Integrand example3	<b>2DITG</b>	2D Integration	Simple
<b>F1XY</b>	Integrand example1	<b>FITG2</b>	2D Integration	Recursive
<b>F2XY</b>	Integrand example2	<b>FITG2</b>	2D Integration	Recursive
<b>F1, F2</b>	Operands example1	<b>FRT2</b>	2-equation System,	Recursive
<b>G1, G2</b>	Operands example2	<b>FRT2</b>	2-equation System	Recursive
<b>CI</b>	Cosine Integral	<b>FINTG</b>	Single Integration	Simple
<b>SI</b>	Sine Integral	<b>FINTG</b>	Single Integration	Simple
<b>ERF</b>	Error Function	<b>FINTG</b>	Single Integration	Simple
<b>FOURN</b>	Fourier Coeffs.	<b>FINTG</b>	Single Integration	Simple
<b>Q-1</b>	Inverse Probability	<b>FROOT, FINTG</b>	Roots of integral	Nested
<b>JNX</b>	Bessel Function J	<b>FINTG</b>	Single Integration	Simple
<b>JITX</b>	Integral Bessel J	<b>FITG2</b>	2D Integration	Recursive
<b>JZRN</b>	Zeros Bessel J	<b>FROOT, FINTG</b>	Roots of integral	Nested

The integrand functions for these examples are direct transcriptions of the well-known definition formulas for the special functions, such as:

$$\text{Si}(x) = \int_0^x \frac{\sin t}{t} dt \quad ; \quad \text{Ci}(x) = \gamma + \ln x + \int_0^x \frac{\cos t - 1}{t} dt$$

$$\text{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt. \quad ; \quad \Phi(x) = \frac{1}{2} \left[ 1 + \text{erf} \left( \frac{x}{\sqrt{2}} \right) \right]$$

For the Inverse Normal probability, the program solves for x using the definition of the cumulative probability based on the error function – which in turn is defined as an integral.

For the Fourier coefficients case, the function is expected to be periodic (but the generic period needn't be  $\pi$ ), and centered around  $x=0$ . The formulas for  $T = \pi$  are given below:

$$a_n = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \cos(nx) dx, \quad n \geq 0$$

$$b_n = \frac{1}{\pi} \int_{-\pi}^{\pi} f(x) \sin(nx) dx, \quad n \geq 1$$

*Example:-* Calculate the first six coefficients for  $F(x) = x^2$ , assuming a period  $T=2\pi$ , centered in  $x_0 = 0$ . As it's known,  $X^2 = 4/3 \pi^2 + \text{SUM}\{ 4 \cos(nx) / n^2 - 4\pi \sin(nx) / n \} | n=0,1,\dots$

Using an accuracy of 6 decimal places:

01 LBL "FT"  
02 X^2  
03 RTN

a0 = 13,1595	b0 = 0
a1 = 4	b1 = -12,566
a2 = 1	b2 = -6,5797
a3 = 0,4444	b3 = -4,1888
a4 = 0,250	b4 = -3,1415
a5 = 0,160	b5 = -2.513

## Other Programs included.

Besides the recursive routines the SIROM module includes a couple of FOCAL programs also geared towards the root finding and integration subjects.

The first program is "**SYS2**", original written by FJ Pamies Durá, (UPLÉ #35006) to solve non-linear systems of 2-equations. Here the operand functions assume x stored in R01 and y stored in R02 for both functions, therefore not entirely compatible with the requirements for the second function in **FRT2** –but it's easy to make a compatible version, just preamble the F2 routine with {RCL 02, ENTER^ } to make it compatible with **SYS2**, i.e:

```
{ LBL "F22, RCL 02, ENTER^ , GTO "F2" } and:
{ LBL "G22, RCL 02, ENTER^, GTO "G2" }
```

This program is noticeable faster than **FRT2** because the iterative process follows a simultaneous approach on both variables (x,y) to approximate for the root, based on the two partial derivatives of the functions respectively. By contrast, on **FRT2** this is done on each variable independently, which in general will require more iterations and therefore longer execution times. It'll be akin to zigzagging versus doing a bee-line towards the root. It also shows the successive values of the approximations in the display:




This program prompts for all the required data up front, in a "driver" arrangement. This includes the two function names, the accuracy (number of decimals), and the initial estimation (x0, y0) for the solution. **SYS2** uses data registers {R00 – R10}.

*Example.* Obtain the roots for the system of two equations below (available as "G1" and "G2")

$$\left\{ \begin{array}{l} g1(x,y) = x^2 + y^2 - 5 \\ g2(x,y) = x^2 - y^2 - 3 \end{array} \right. \quad \text{Solution:} \quad \begin{array}{l} x = 2 \\ y = 1 \end{array}$$

This is an interesting case because FRT2 not only is much slower (as we knew it was going to be), but also fails to find a root using initial guesses equal to the solutions, i.e. x0 = 2, y0=1.





The second program is "**2DITG**" – Valentín Albillo's neat example from DataFile for Double Integrals - that uses **FINTG** applied to the inner integral and with a 3-point, 5<sup>th</sup> order Gaussian method for the outer integral. Three example routines (2DF1, 2DF2, 2DF3) are included in the ROM as follows:

$$I = \int_0^1 \int_1^2 (x^2 + y^2) \cdot dy \cdot dx \quad ; \quad I = \int_3^4 \int_1^2 1/(x + y)^2 \cdot dy \cdot dx$$

$$I = \int_{-2.3}^{1.6} \int_{3.9}^{6.1} (e^{-x*x} + x^3 - y^3 * x^2 + 7) * \tan^{-1}(x-2) * \sin(y+3) \cdot dy \cdot dx$$

See the original article for details, available at:

<http://web.archive.org/web/20110906135412/http://membres.multimania.fr/albillo/calc/pdf/DatafileVA024.pdf>


In addition to the outer and inner limits in the stack (lower outer first, upper inner last) this program uses a parameter to specify the number of steps for the outer integration (a.k.a. the number of subintervals used). This is expected to be in R10 before executing the program – make sure you provide the value prior to executing 2DITG. The accuracy and execution times are directly impacted by the display settings (by virtue of FINTG) and the number of steps.

If you want to use **2DITG** with your own functions, the integrand routines expect x in R01 and y in the X register (as per **FINTG** conventions).

The results are:      I1 = 8/3 = 2.6666666  
                              I2 = Ln(25/24) = 0.040821  
                              I3 = 1321.275779

Note that because the data registers convention is the same for both main programs, you can use 2DF1, 2DF2 and 2DF3 as examples for **FITG2** – and conversely use F1XY and F2XY as examples for **2DITG**. In general, the execution times are very similar (and long) using M=10 for number of steps.

3D plot	$\frac{1}{(x+y)^2}$
---------	---------------------





Finally, there is the PPC Solve routine, a classic must-have in everybody's toolbox. It is labeled "**SLV**" here, and the only difference is that it expects the function name in Alpha (no need to save it in R06 before). **SLV** uses data registers {R06-R09}. Set F10 if you want to see the iterations. Refer to the PPC ROM manual for details on SV, which also includes numerous examples of utilization.

### Other References.

There was no more room available in this module to include more applications or examples, but many more fancy programs to calculate single and multiple integrals are available in the **INTEGRATOR** module, written in collaboration with Jean-marc Baillard and including a small arsenal of tools for the task. And yet more integration programs are available in the **SwapMath** module, written by Sherman Lowell and Ernest Gibbs and published in PPCCJ V11N7 p18 and PPCCJ V8N4 p31 respectively – you're encouraged to refer to their QRG's for further details.

If what suits your fancy is systems of non-linear equations instead, you're encouraged to check the **NONLINEAR** module, which picks up where this module leaves off, including JM Baillard's programs to resolve systems with two, three and general case with "n" equations.

